Maple: Решаване на обикновени диференциални уравнения

Въведение

Диференциално уравнение е уравнение, в което участват производни на една или повече неизвестни функции. Решаването на диференциалното уравнение означава да се намери функция (или всички функции), която удовлетворява диференциалното уравнение. Много от основните закони на физиката, химията, биологията и икономиката могат да се формулират с диференциални уравнения. Диференциалните уравнения често се класифицират по отношение на техния ред. Редът на диференциалното уравнение е редът на най-високата производна, участваща в него. Обикновеното диференциално уравнение (ОДУ) е диференциално уравнение, в което търсената неизвестна функция е функция на една независима променлива. По-долу ще разглеждаме само ОДУ от първи и втори ред.

Получаване на общо решение на ОДУ от първи ред

Ще решаваме следното ОДУ от първи ред,

$$\frac{dy}{dx} = 2xy, \qquad y(0) = 2. \tag{1}$$

Това по-скоро е директно решимо ОДУ, но то ще ни позволи да демонстрираме начина за решаване на ОДУ, използвайки Maple. Да отбележим че уравнението (1) има начално условие. Това означава, че ще се опитаме да намерим точно решение, зависещо от частното начално условие, в този случай $y(0) = 2$: стойността на y, когато x е 0, е 2. Първо ще решим уравнението, игнорирайки началните условия и ще получим общо решение. Преди да се опитаме да решаваме ОДУ в Maple е необходимо да „заредим" частните команди, и функции, които ще са ни нужни. Командите по отношение на ОДУ в Maple са обединени в пакета “DEtools”. Допълнителните команди за изчертаване на кривите на решенията на ОДУ могат да бъдат намерени в пакета “plots”. Следователно, нашата първа стъпка е да заредим тези два пакета, използвайки командата with:

```
> with(plots):
> with(DEtools):
```

Следващата стъпка е да въведем ОДУ, което искаме да решим. Ще припомним, че функцията *y* зависи от *x* и за това е достатъчно да въведем *y*(*x*), така че Maple да е в състояние да разпознае зависимостта. Ще отбележим уравнението (1) като ODE1 използвайки оператора за присвояване:

```
> ODE1:=diff(y(x),x)=2*x*y(x);
```

$$ODE1 := \frac{d}{dx}\mathrm{y}(x) = 2\,x\,\mathrm{y}(x)$$

Командата за решаване но ОДУ е dsolve. Да припомним, че ако не познавате някоя команда, може да извикате помощния файл за нея, чрез добавяне на въпросителен знак (?) пред командата и натискане на **Enter**. Сега ще решим уравнение (1) и ще получим общото му решение.

```
> dsolve(ODE1,y(x));
```

$$\mathrm{y}(x) = \_C1\, e^{(x^2)}$$

Това е общото решение на уравнение (1). Да отбележим, че “$\_C1$” е начина на Maple да представи произволна константа. В по-сложни решения тази произволна константа може да се появи след члена, с който е свързана.

Също така е възможно да начертаем графиките на кривите на общото решение. За да видим графиките на кривите със съответното поле на посоките на общото решение на уравнение (1), използваме следната команда:

```
> dfieldplot(ODE1,y(x),x=-2..2,y=-2..2,color=blue,scaling=constrained,
> dirgrid=[40,40]);
```



Първият елемент в скобите е диференциалното уравнение ODE1. Вторият аргумент е името на зависимата променлива $y(x)$. Третият и четвъртият дават интервалите на независимата и зависимата променливи $x=-2..2,\ y=-2..2$. Останалите елементи са опции и могат да бъдат изпуснати по наше желание. Обща практика е винаги, когато се изчертава направление на полета да се използва опцията “scaling = constrained” тъй като в противен случай графиката ще изглежда неточна, защото направляващите линии ще са изкривени.

Получаване на частно решение на ОДУ от първи ред

Когато е дадено начално условие на задачата, с Maple можем да определим съответното частно решение на общото решение, което удовлетворява това начално условие. Решаваме отново ОДУ с дадено в (1) начално условие, използвайки командата dsolve.

```
> dsolve({ODE1,y(0)=2},y(x));
```

$$y(x)=2\,e^{(x^2)}$$

Ако искаме да начертаем графиката на кривата на частното решение, трябва да използваме командата DEplot. DEplot ще начертае кривите на насоченото поле така, както линиите на частното решение зависят от началното условие.

```
> DEplot(ODE1,y(x),-2..2,[y(0)=2],linecolor=magenta,color=blue,
> arrows=LINE);
```



Ако не искаме да изчертаваме направлението на полето с частното решение, използваме опцията "arrows=NONE" в командата DEplot.

Решаване на ОДУ от втори ред с Maple

Ще започнем с решаването на следното хомогенно уравнение от втори ред с постоянни коефициенти:

$$y'' + 2y' + 10y = 0.$$

Това уравнение е хомогенно, защото всички членове, в които участва неизвестната функция *y* и нейните производни се срещат от лявата страна на уравнението, а дясната страна е нула. Ще започнем въвеждането на горното уравнение в Maple. Да припомним, че също изискваме пакетите “DEtools” и “plots” за решаване на ОДУ от втори ред и трябва да ги добавим както при примера с ОДУ от първи ред.

```
> eq1 := diff(y(t),t,t) + 2*diff(y(t),t) + 10*y(t) = 0;
```

$$eq1 := \left(\frac{d^2}{dt^2}\,y(t)\right) + 2\left(\frac{d}{dt}\,y(t)\right) + 10\,y(t) = 0$$

Да припомним, че diff(y(t),t,t) ще даде втората производна на $y(t)$. Отново ще използваме командата dsolve за решаване на диференциалното уравнение. Първият аргумент е диференциалното уравнение, което решаваме (eq1), а втория е функцията, която търсим (y(t)). Да припомним, че в предния пример, когато използвахме командата dsolve тя ни дава решението във формата “неизвестна променлива = решение”. Ако искаме да видим само решението, може да използваме командата rhs за извеждане на дисплея само на дясната страна. Ще запомним решението в променливата “sol1”.

```
> sol1 := rhs(dsolve(eq1,y(t)));
```

$$sol1 := \_C1\,e^{(-t)}\sin(3\,t) + \_C2\,e^{(-t)}\cos(3\,t)$$

Да припомним че с “$\_C1$” и “$\_C2$” Maple представя произволните константи.

Решаване на ОДУ от втори ред с начални условия

Следващата стъпка ще бъде да решаването на ОДУ от втори ред със зададени начални условия. Във всяко ОДУ от втори ред може да има две начални условия. Сега ще решим първото ОДУ от втори ред при следните начални условия:

$$y'' + 2y' + 10y = 0, \qquad y(0) = 3, y'(0) = -5.$$

Използваме командата dsolve за да решим ОДУ, удовлетворяващо дадените начални условия. Записваме решението в променливата sol2.

```
> sol2 := rhs(dsolve({eq1,y(0)=3,D(y)(0)=-5},y(t)));
```

$$sol2 := -\frac{2}{3}\,e^{(-t)}\sin(3\,t) + 3\,e^{(-t)}\cos(3\,t)$$

Сега имаме точното (или частното) решение на уравнението. Начертавайки графиката решението, по-лесно ще разберем за какво става дума. Просто използваме командата plot за графиката на sol2.

```
> plot(sol2,t=-1..6,labels=["t","y"]);
```



Нехомогенни ОДУ от втори ред с начални условия

Ще разгледаме как се решават нехомогенни ОДУ от втори ред с начални условия. Да разгледаме следната задача:

$$y'' + y' + y = t^2\cos(2t), \qquad y(0) = 0, y'(0) = 2.$$

Както преди, нашата първа стъпка е да въведем уравнението в Maple, игнорирайки началните условия. Ще означим новото ОДУ като eq2.

```
> eq2 := diff(y(t),t,t) + diff(y(t),t) + y(t) = t^2*cos(2*t);
```

$$eq2 := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + \left(\frac{d}{dt}\,\mathrm{y}(t)\right) + \mathrm{y}(t) = t^2\cos(2\,t)$$

Отново можем да решим това ОДУ от втори ред, използвайки командата dsolve. Ще означим решението със sol3.

```
> sol3 := rhs(dsolve(eq2,y(t)));
```

$$sol3 := e^{\left(-\frac{t}{2}\right)} \sin\left(\frac{\sqrt{3}\,t}{2}\right) \_C2 + e^{\left(-\frac{t}{2}\right)} \cos\left(\frac{\sqrt{3}\,t}{2}\right) \_C1 + \frac{1}{2197} (338\,t^2 + 832\,t - 1212) \sin(2\,t)$$
$$+ \frac{1}{2197} \cos(2\,t)\,(336 - 507\,t^2 + 1118\,t)$$

Това решение е по-сложно от предишния пример, поради нехомогенните членове в дясната страна на уравнението. Сега ще решим задачата, като вземем впредвид началните условия. Отново това може да се направи лесно с командата dsolve. Означаваме решението с началните условия със sol4.

```
> sol4 := rhs(dsolve({eq2,y(0)=0,D(y)(0)=2},y(t)));
```

$$sol4 := \frac{3688}{2197} e^{\left(-\frac{t}{2}\right)} \sin\left(\frac{\sqrt{3}\,t}{2}\right) \sqrt{3} - \frac{336}{2197} e^{\left(-\frac{t}{2}\right)} \cos\left(\frac{\sqrt{3}\,t}{2}\right)$$
$$+ \frac{1}{2197} (338\,t^2 + 832\,t - 1212) \sin(2\,t) + \frac{1}{2197} \cos(2\,t)\,(336 - 507\,t^2 + 1118\,t)$$

Дори да се вземат предвид началните условия, този отговор е доста сложен, така че начертаването на горната функция може да ни помогне да видим цялостното поведение с графиката на решението.

```
> plot(sol4,t=0..18,labels=["t","y"]);
```